PharmaSUG 2022 - Paper SA-004

A SAS macro wrapper for an efficient Deming regression algorithm via PROC IML: The Wicklin Method

Jesse A. Canchola and Natasha Oza, Roche Diagnostics Solutions, Pleasanton, CA, USA
Ben Wang, BioMarin Pharmaceutical, Inc., San Rafael, CA, USA

ABSTRACT

Deming regression is one tool in method comparison/correlation studies and is used to compare two or more quantitative measurements from two similar methods that produce like-measurements on the same subject/sample on the linear scale where both methods can be deemed to have error (Deming, 1943). Contrast this with linear regression (viz., Y=mX+b+error) where only the dependent variable Y is deemed to have error and the X variable is fixed (i.e., without error, for example, an expected value). At least two published SAS macros exist at this time that perform Deming regression (Deal, Pate, and El Rouby in 2009 and Njoya and Hemyari in 2017). However, these macros can lengthen the time to final result as they both use a macro loop to perform the jackknife or bootstrap. The method proposed by Wicklin (2019) is deemed as substantially more efficient as it uses SAS/IML with explicit formulas to compute the slope and intercept of the Deming regression line. We introduce a SAS macro wrapper for the Wicklin method that produces camera-ready graphs of the Deming regression line drawn over the scatterplot of measurement pairs with additional enhancements.

INTRODUCTION

DEMREG is a SAS macro for obtaining Deming regression parameter estimates, associated 95% confidence intervals, an x-y scatter plot overlaid with the Deming regression line, confidence bands (loosely assuming a reference for visually capturing deviant observations) and the Deming regression equation and additional model parameter estimates on the graphical plot. The macro uses the SAS/IML as the main computation engine and the SGPLOT procedure as the main graphing tool.

DEMING REGRESSION THEORY

Measurement/observational error is the difference between a measured quantity and its value taken as “true”. There are at least three types of measurement/observational errors: random, systematic, and proportional. Random errors are “naturally” or “typically” occurring errors that are to be expected. On the other hand, systematic errors are errors that are consistent for the majority of measurements as compared to the true values (e.g., errors produced by a mis-calibrated instrument) (Glen, 2022). Proportional measurement/observational errors are those errors produced in an increasing or decreasing proportion over the length of the data range as compared to the true values.

As in any linear regression, we assume a linear relationship between the measurements of both methods to be compared. If your data are not linear, consider applying a proper transformation to linearity (LaLonde SM, 2005; Steiger JH, 2009). For example, with exponential data [as in molecular diagnostics due to the nature of the polymerase chain reaction (PCR) exponential response], a proper transform would be the logarithmic transform (e.g., $\log_e(\bullet)$, $\log_{10}(\bullet)$, etc.)

The general Deming regression theory has been extensively discussed. The theory was originally introduced by Adcock (1878) for the case lambda (λ), the ratio of the total error variances of X and Y, equal to 1. Kummell in 1879 generalized this to include any defined λ. Koopmans in 1936 and Deming in 1943 are credited with expanding the theory for wider use with Deming eventually

having his name tied to the theory because of his popular book on the subject (Wikipedia, 2022). Refer to Deming (1943) and Njoya and Hemyari (2017) for an expanded discussion of the theory.

The reader should note that in this macro, when lambda (λ) is not specified, we follow the definition of λ due to Linnet (1990) as S_{xx}/S_{yy} rather than S_{yy}/S_{xx} as other authors have defined (Miller et al., 1980; Kendall and Stuart, 1961; Francq and Govaerts, 2014; Hall, 2022). Additionally, we do not specify a Pearson correlation coefficient that is well defined within the ordinary linear regression theory framework but not within the methodology that underpins Deming regression (i.e., errors-in-variables, orthogonal regression, measurement error models, regression with errors in X, etc.)

PROGRAMMING

The SAS/IML code provided by Wicklin (2019) is modified slightly in order to accommodate the enhancements as discussed here. The entire source code set is shown in Appendix A. The example macro call is included in Appendix B. For convenience, the user may download the macro and supporting documents and any updates at this online link: https://bit.ly/3uVlvXm

METHODS

There are four main steps to successfully run the DEMREG SAS macro as follows.
Step 1: Prepare your data paying attention to the required macro variable inputs from Table 1, below.

a. Input data name ❶
b. Output data name ❷
c. Unique sample identifier ❺
d. "old method" or reference or "independent" or "X" variable ❻
e. "new method" or test or "dependent" or "Y" variable ❽

Note: The macro will automatically keep only valid paired data and will exclude any missing data using pairwise deletion.

Step 2: Load the DEMREG.sas SAS macro into your session.

%include "c:/sasmacros/DEMREG.sas" ;

Step 3: Enter DEMREG SAS macro inputs (see later example and **Appendix B**).

```
ods noproctitle ;
ods rtf file = "&path./results/rtf/DemReg_Example_&sysdate..rtf" ;
%demreg(dsin=
,  dsout=
,  title1= "Your title 1"
,  title2= "Your sub-title 2"
,  sampleid=
,  xvar=
,  xlab=
,  yvar=
,  ylab=
,  alpha= 0.05
,  lambda= 1
,  GraphMin=
,  GraphMax=
,  GraphIncrement= 1.0
,  ParmDecimalPlaces= 0.001) ;
ods rtf close ;
```

Step 3: Run your SAS code.

Parameter Number	Macro Parameter	Parameter Description	Specifications / Defaults
❶	`dsin`	SAS input data set	required
❷	`dsout`	SAS output data set	required
❸	`title1`	Main title	optional
❹	`title2`	Subtitle	optional
❺	`sampleid`	Unique row observation identifier	required
❻	`xvar`	"reference" or "independent variable	required
❼	`xlab`	Label for `xvar`	optional
❽	`yvar`	"test" or "dependent" variable	required
❾	`ylab`	Label for `ylab`	optional
❿	`alpha`	Significance level	optional / 0.05
⓫	`lambda`	lambda is the ratio of the variances (or standard deviations). If lambda=1, then the variances are assumed equal.	optional / 1.0 NOTE: if lambda is not specified, it is estimated as S_{xx}/S_{yy} (following Linnet; 1990, 1993).
⓬	`GraphMin`	Minimum value for output graph	optional / SGPLOT auto-determined
⓭	`GraphMax`	Maximum value for output graph	optional / SGPLOT auto-determined
⓮	`GraphIncrement`	Increment on x- and y-axes	optional / 1.0
⓯	`ParmDecimalPlaces`	Decimal places for outputted values	optional / 0.001

Table 1. DEMREG macro variable inputs.

EXAMPLES

Next, an example shows the execution of the DEMREG SAS macro. Deal, Pate, and El Rouby (2009) generate a data set as follows. The authors have added fixed randomization seeds in order to produce the same data set each time the code is run. The full code for this example can be found in **Appendix B**:

```
%LET SEED1 = 726453819 ;
%LET SEED2 = 289546394 ;
%LET SEED3 = 923847362 ;
ods listing close;
data Example ;
 do MyId = 1 to 200 ;
  Value = UNIFORM(&SEED1.) * 10 ;
  Ref = Value + NORMAL(&SEED2.) ;
  Test = Value + NORMAL(&SEED3.) ;
  output ;
 end ;
 drop Value ;
run ;
```

A sample screenshot of a portion of the resulting 200 observation data generated (using the given randomization SEEDs) looks like this:

MyId	Ref	Test
1	3.60476387...	1.92649262...
2	3.76138859...	3.16204556...
3	5.76213514...	5.889339305
4	2.61637987...	3.18489284...
5	2.58974104...	3.043169389
6	2.49173080...	4.80926629...

where the variables are:

MyID = unique observation identifier;

Ref = Reference or "independent" variable (X);

Test = Test or "dependent" variable (Y).

The results below include the parameter estimates superimposed on the scatterplot overlaid with the Deming regression line.



Figure 1. the DEMREG SAS macro-produced scatterplot of *test* vs. *reference* overlaid with the Deming regression line (solid blue) using lambda=1 (see **Appendix B**). These are the same data in both panels. The panel on the left includes the 95% mean (solid red) and individual (solid teal) confidence bands whilst the right panel does not. Both graphs contain the 45-degree line from which to measure "perfect" agreement (dotted black) against the Deming line (solid blue). Also included are the sample size, Deming regression slope, intercept and corresponding 95% confidence intervals.

DISCUSSION

We have seen that the DEMREG SAS macro is simple to use with the efficiency gains using the Wicklin method via PROC IML. These efficiency gains are apparent even with the smallest of data sets. Table 2 shows the comparison of the **Example 1** results from the three available macros using lambda (λ) = 1. The estimates all agree to a certain extent.

Deming Macro	Method for Variance Estimation	Slope Estimate (95% CI)	Intercept Estimate (95% CI)
Canchola, Oza & Wang (2022) ($\lambda=S_{xx}/S_{yy}$)	Explicit Formulas Due to Wicklin (2019)	0.936 (0.853, 1.019)	0.389 (-0.050, 0.828)
Canchola, Oza & Wang (2022) (λ=1)	Explicit Formulas Due to Wicklin (2019)	0.943 (0.868, 1.017)	0.355 (-0.047, 0.757)
Njoya & Hemyari (2017) (λ=1)	Bootstrap (10000 samples)	0.943 (0.873, 1.021)	0.350 (-0.053, 0.745)
Deal, Pate and Rouby (2009) (λ=1)	Jacknife	0.943 (0.868, 1.017)	0.355 (-0.045, 0.755)

Table 2. Comparison of Deming regression estimates from three different SAS macros, including the current one for λ=1 and $\lambda=S_{xx}/S_{yy}$ (following Linnet; 1990).

Note that for obtaining the Njoya & Hemyari (2017) macro estimates for this data set example, we used the following code (with adjustments made in **bold**):

```
/* Re-sampling to create the input datasets for the bootstrap */
proc surveyselect
  data=&datain out=resampled
  seed=&seedBoot
  method=urs
  samprate=1
  outhits
  reps=&num_rep ;
  id _all_ ;
run ;
```

Finally, the various graphical representations of the same X-Y relationship outputted by the DEMREG SAS macro give the reader an idea of which types of graphs are desired and allow further customization in a “pinch”.

CONCLUSION

The DEMREG SAS macro can be used to produce camera-ready graphs of the Deming regression line drawn over the scatterplot of paired measurements from data generators not considered to be without error. Additional enhancements include 95% confidence bands for the individual observations as well as for the mean values across the linear range of the measurements.

REFERENCES

1. Adcock, R. J. (1878). "A problem in least squares". The Analyst. 5 (2): 53–54. doi:10.2307/2635758. JSTOR 2635758.
2. Deal AM, Pate VW, El Rouby S (2009). "A SAS® Macro for Deming Regression". Proceedings of the PharmaSUG 2009 Conference. Portland, OR. Paper CC-014.
3. Deming WE (1943). Statistical adjustment of data. Wiley.
4. Francq BG, Govaerts BB (2014). "Measurement methods comparison with errors-in-variables regressions. From horizontal to vertical OLS regression, review and new perspectives." Chemometrics and Intelligent Laboratory Systems, 134, 123-139. (https://doi.org/10.1016/j.chemolab.2014.03.006).
5. Glen S (2022). "Measurement Error (Observational Error)" From StatisticsHowTo.com: Elementary Statistics for the rest of us! https://www.statisticshowto.com/measurement-error/ accessed on 28Feb2022.
6. Hall J (2022). "Linear Deming Regression" (https://www.mathworks.com/matlabcentral/fileexchange/33484-linear-deming-regression), MATLAB Central File Exchange. Retrieved March 29, 2022.
7. Kendall MG and Stuart A (1961). The advanced theory of statistics. London: Griffin, Vol. II, Ch. 29, pp. 357-418.
8. Koopmans, T. C. (1936). Linear regression analysis of economic time series. DeErven F. Bohn, Haarlem, Netherlands.
9. Kummell, C. H. (1879). "Reduction of observation equations which contain more than one observed quantity". The Analyst. 6 (4): 97–105. doi:10.2307/2635646. JSTOR 2635646.
10. LaLonde SM (2005). "Transforming Variables for Normality and Linearity – When, How, Why and Why Not's". Proceedings of NESUG 18 held in Portland, ME.
11. Linnet K (1990). "Estimation of the linear relationship between the measurements of two methods with proportional errors." Statistics in Medicine, 9, 1463-1473.
12. Linnet K (1993). "Evaluation of Regression Procedures for Methods Comparison Studies." CLIN.CHEM, 39, 3, 424-432.
13. Miller RG, Efron B, Brown Jr. BW, Moses LE (1980). Biostatistics casebook. John Wiley & Sons, New York, 1980. 238 pp.
14. Njoya M and Hemyari P (2017). "Application of Deming Regression in Molecular Diagnostics using a SAS® Macro." Proceedings of the PharmaSUG 2017 Conference. Baltimore, MD. Paper TT12.
15. Steiger JH (2009). "Transforming to linearity." Department of Psychology and Human Development Vanderbilt University. http://statpower.net/Content/313/Lecture%20Notes/Transformation.pdf accessed on 20Mar2022.
16. Wicklin R (2019). "Deming regression for comparing different measurement methods". SAS Blogs. website: https://blogs.sas.com/content/iml/2019/01/07/deming-regression-sas.html accessed on 07Nov2021.
17. Wikipedia. 2022. "Deming Regression." Last modified 10-Feb-2022. https://en.wikipedia.org/wiki/Deming_regression.

ACKNOWLEDGMENTS

The authors thank the Roche reviewers and Alison J. Canchola for their valuable input.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the corresponding author at:

Jesse A. Canchola
Roche Molecular Systems, Inc.
4300 Hacienda Drive
Pleasanton, CA 94588
E-mail 1: Jesse.Canchola@Roche.com
E-mail 2: JCanch@Gmail.com

The user may download the macro and supporting documents and any updates at this online link: https://bit.ly/3uVlvXm

APPENDIX A: DEMREG SAS MACRO SOURCE CODE

*** BEGIN MACRO CODE ;**

```
* ******************************************************************** ;
* SAS Macro DEMREG
*
* This macro produces the following:
* 1. Estimates of Deming regression parameters and associated 95% CIs
* 2. Various scatterplots with Deming regression line overlays
* 3. Confidence bands are included at the stated confidence level:
*   (1-alpha)*100% where alpha = significance level default set at 0.05
*
* Macro Inputs:
*
* dsin: enter input data set name
* dsout: enter output data set name
* title1: enter your main title
* title2: enter your subtitle
* sampleid: enter unique observation id
* xvar: enter "reference" or "independent" variable
* xlab: enter label for xvar
* yvar: enter "test" or "dependent" variable
* ylab: enter label for yvar
* alpha: enter significance level
* GraphMin: enter minimum value for output graph
* GraphMax: enter maximum value for output graph
* GraphIncrement: enter increment on x- and y-axes
* ParmDecimalPlaces: enter decimal places for outputted values
*
* Macro Defaults
* %macro demreg(dsin=              /* required */
* ,     dsout=                     /* required */
* ,     title1=                    /* optional */
* ,     title2=                    /* optional */
* ,     sampleid=                  /* required */
* ,     xvar=                      /* required */
* ,     xlab= "Reference"          /* optional */
* ,     yvar=                      /* required */
* ,     ylab= "Test"               /* optional */
* ,     alpha= 0.05                /* optional */
* ,     lambda= 1                   /* optional */
* ,     GraphMin=                  /* optional */
* ,     GraphMax=                  /* optional */
* ,     GraphIncrement= 1          /* optional */
* ,     ParmDecimalPlaces= 0.001) ; /* optional */
*
* Main PROC IML engine source code: Rick Wicklin [Ref. Wicklin R (2019).
*  "Deming regression for comparing different measurement methods".
*  SAS Blogs. website:  https://blogs.sas.com/content/iml/2019/01/07/deming-regression-sas.html
*  accessed on 07Nov2021]
*
* Macro Wrapper Authors: Jesse A. Canchola (JAC), Ben Wang (BW)
* Creation Date: 20Mar2022
*
* Validator(s):   Natasha Oza (NA) and JAC
* Validation Date: 20Mar2022
*
* ******************************************************************** ;

%macro demreg(dsin=
,       dsout=
,       title1=
,       title2=
,       sampleid=
,       xvar=
,       xlab= "Reference Label"
,       yvar=
,       ylab= "Test Label"
,       alpha= 0.05
,       lambda= 1
,       GraphMin=
```

```
,       GraphMax=
,       GraphIncrement= 1
,       ParmDecimalPlaces= 0.001) ;

* Defaults Section ;
%if &alpha.             = %then %let alpha = 0.05 ;
%if &ParmDecimalPlaces. = %then %let ParmDecimalPlaces = 0.001 ;
%if &GraphIncrement.    = %then %let GraphIncrement = 1 ;
* min and max of data for axis defaults other than SAS defaults
* if some or none of graphical parms are specified in macro ;
* variables created by proc sql are automatically macro variables
* when using "into" with a colon in front rather than "as" with no colon ;
proc sql noprint ;
  select min(&xvar.) into :xvar_min
  from &dsin. ;
quit;
proc sql noprint ;
  select max(&xvar.) into :xvar_max
  from &dsin. ;
quit;
%if &GraphMin. = %then %let GraphMin1 = &xvar_min. ;
%if &GraphMax. = %then %let GraphMax1 = &xvar_max. ;
* keep only valid complete pairs ;
proc sort data=&dsin. ; by &sampleid. ; run ;
data complete ;
  set &dsin. ;
  diff = &yvar. - &xvar. ;
  if diff ne . ;
  drop diff ;
run ;

* Wicklin MODIFIED Code ;
proc iml ;
start Deming(XY, lambda=&lambda.) ;
   * Equations from https://en.wikipedia.org/wiki/Deming_regression ;
   m = mean(XY) ;
   xMean = m[1] ; yMean = m[2] ;
   S = cov(XY) ;
   Sxx = S[1,1] ; Sxy = S[1,2] ; Syy = S[2,2] ;
   * if lambda is specified (e.g., lambda=1), use it. Otherwise, estimate. ;
   if IsEmpty(lambda) then
      delta = Sxx / Syy ;        * estimate of ratio of variance ;
   else delta = lambda ;
   c = Syy - delta*Sxx ;
   b1 = (c + sqrt(c**2 + 4*delta*Sxy**2)) / (2*Sxy) ;
   b0 = yMean - b1*xMean ;
   return (b0 || b1) ;
finish ;

use complete ; read all var {&xvar. &yvar.} into XY ; close ;
b = Deming(XY,&lambda.) ;
* first check of point estimates ;
print b[c={'Intercept' 'Slope'} L="Deming Regression"] ;

* Helper modules for jackknife estimates of standard error and CI for parameters: ;
* return the vector {1,2,...,i-1, i+1,...,n}, which excludes the scalar value i ;
start SeqExclude(n,i) ;
   if i=1 then return 2:n ;
   if i=n then return 1:n-1 ;
   return (1:i-1) || (i+1:n) ;
finish ;

* return the i_th jackknife sample for (n x p) matrix X ;
start JackSamp(X,i) ;
   return X[ SeqExclude(nrow(X), i), ] ;  * return data without i_th row ;
finish ;

* 1. Compute T = statistic on original data ;
T = b ;

* 2. Compute statistic on each leave-one-out jackknife sample ;
```

```
n = nrow(XY) ;
call symputx("n",n) ;
T_LOO = j(n,2,.) ;              * LOO = "Leave One Out" ;
do i = 1 to n ;
   J = JackSamp(XY,i) ;
   T_LOO[i,] = Deming(J) ;
end ;

* 3. compute mean of the LOO statistics ;
T_Avg = mean( T_LOO ) ;

* 4. Compute jackknife estimates of standard error and CI ;
stdErrJack = sqrt( (n-1)/n * (T_LOO - T_Avg)[##,] ) ;
alpha = 0.05 ;
tinv = quantile("T", 1-alpha/2, n-2) ; /* use df=n-2 b/c both x and y are estimated */
Lower = T - tinv#stdErrJack ;
Upper = T + tinv#stdErrJack ;
result = T` || T_Avg` || stdErrJack` || Lower` || Upper` ;

create ParmDat var {T T_Avg stdErrJack lower upper} ; * create data set ;
append ;          * write data in vectors ;
close ParmDat ; * close the data set ;

* final check of point estimates and associated 95% CIs ;
print result[c={"Estimate" "Mean Jackknife Estimate" "Std Error" "Lower 95% CL" "Upper 95% CL"}
r={'Intercept' 'Slope'}] ;
run ;
* END WICKLIN MODIFIED CODE ;
data ParmData ;
  set ParmDat ;
  obsno=_n_ ;

  if obsno = 1 then
     do ;
             Intercept = T ;
                InterceptCI = compress("(" || round(Lower,&ParmDecimalPlaces.) || ", " ||
round(Upper,&ParmDecimalPlaces.) || ")") ;
         call symputx("Intercept",round(Intercept,&ParmDecimalPlaces.)) ;
                call symputx("InterceptCI",InterceptCI) ;
         end ;

  if obsno = 2 then
     do ;
             Slope = T ;
                SlopeCI = compress("(" || round(Lower,&ParmDecimalPlaces.) || ", " ||
round(Upper,&ParmDecimalPlaces.) || ")") ;
         call symputx("Slope",round(Slope,&ParmDecimalPlaces.)) ;
                call symputx("SlopeCI",SlopeCI) ;
         end ;

run ;

proc iml ;
use complete ; * read data ;
read all var {&xvar.} into x ;
read all var {&yvar.} into y ;
close complete ;

* 1a. Find min and max of data ;
xMin = min(x) ; xMax = max(x) ;
yMin = min(y) ; yMax = max(y) ;
* 1b. Find mean ;
xbar = mean(x) ;

* 2. Store as macro variables ;
call symput("xMin", char(xMin)) ;
call symput("xMax", char(xMax)) ;
call symput("yMin", char(yMin)) ;
call symput("yMax", char(yMax)) ;
call symput("xbar", char(xbar)) ;
```

```
* Linear helper function ;
start L(x) ;
   a = &intercept ; * intercept ;
   b = &slope ; * slope ;
   return(a + b*x) ;
finish ;

* 3. Store the origin in macro variables ;
call symput("x0", char(xMin)) ;
call symput("y0", char(L(xMin))) ;

* 4. Write vars LX and LY to EndLine data set ;
Lx = xMax ; Ly = L(xMax) ;
create EndLine var {Lx Ly} ;
   append ;
close EndLine ;
run ;

* 5. Calculating confidence and prediction bands ;
data band1 ;
       set complete ;
       xo=&xvar ;
       xbar=&xbar ;
       xdiff=xo-xbar ;
       xdiff2=xdiff**2 ;
run ;
proc summary data=band1 ;
var xdiff2 ;
output out=sxx sum=sxx ;
RUN ;
data _null_ ;
       set sxx ;
       call symput("sxx", sxx) ;
run ;

data band2 ;
       set band1 ;
       sxx=&sxx ;
       tval=tinv(&alpha/2,&n-2) ;

       yhat= &intercept + &slope*xo ;
       yo=&yvar ;
       ydiff=yo - yhat ;
       ydiff2=ydiff**2 ;
run ;
proc summary data=band2 ;
var ydiff2 ;
output out=ssq sum=ssq ;
RUN ;
data _null_ ;
       set ssq ;
       call symput("ssq", ssq) ;
run ;

data band3 ;
       set band2 ;
       s=sqrt(&ssq/(&n-2)) ;
       delta=tval*s*sqrt((1/&n)+(xdiff2/sxx)) ;

       UCL=yhat+delta ;
       LCL=yhat-delta ;

       deltap=tval*s*sqrt(1+(1/&n)+(xdiff2/sxx)) ;
       UCLp=yhat+deltap ;
       LCLp=yhat-deltap ;
run ;

* 6. Final dataset and SGPLOT ;
data final ;
       merge complete band3 ;
       by &sampleid. ;
```

```
run ;
data &dsout. ;
        set final EndLine ;
run ;

proc sgplot data = &dsout. noautolegend ;

        Band X=&xvar lower=LCLp upper=UCLp / nofill
                  lineattrs=(THICKNESS=0.1 COLOR=bibg PATTERN=2) ;
        Band X=&xvar lower=LCL upper=UCL / nofill
                  lineattrs=(THICKNESS=0.1 COLOR=rose PATTERN=2) ;

        scatter y=&yvar x=&xvar / MARKERATTRS=(SIZE=5px)
                  name="Assay1" ; *COLOR=blue SYMBOL=triangle) ;

        vector x=Lx y=Ly /xorigin=&x0 yorigin=&y0 noarrowheads
                                    LINEATTRS = (THICKNESS=0.1 COLOR=Blue PATTERN=1) ;
        LINEPARM x=0 y=0 slope=1 / LEGENDLABEL = 'Unity: Y=X'
                  LINEATTRS = (THICKNESS=2 COLOR=Black PATTERN=2)
                  name="Unity" ; * works in SAS v9.3 ;

%if &GraphMin. eq  and &GraphMax. eq
       %then %do ;
                  YAXIS LABEL = &ylab. ;
                  XAXIS LABEL = &xlab. ;
          %end ;
        %else %do ;
                  YAXIS LABEL = &ylab.
                    VALUES = (
                              %if &GraphMin. = %then
                                  %do ;
                                    &GraphMin1.
                                  %end ;
                               %else %do ;
                                     &GraphMin.
                                %end ;
                                TO
                               %if &GraphMax. = %then
                                   %do ;
                                      &GraphMax1.
                                   %end ;
                                   %else %do ;
                                      &GraphMax.
                                   %end ;
                                   BY &GraphIncrement.) ;
                  XAXIS LABEL = &xlab.
                    VALUES = (
                              %if &GraphMin. = %then
                                  %do ;
                                      &GraphMin1.
                                  %end ;
                                  %else %do ;
                                      &GraphMin.
                                  %end ;
                                  TO
                              %if &GraphMax. = %then
                                  %do ;
                                      &GraphMax1.
                                  %end ;
                                  %else %do ;
                                      &GraphMax.
                                  %end ;
                                  BY &GraphIncrement.
                              ) ;
    %end ;

    INSET "Deming Regression (N=&n): Y = &intercept + &slope X"
        "95% CIs: Intercept=&interceptci ; Slope=&slopeci" / POSITION = BOTTOMRIGHT BORDER ;

        TITLE1 &title1 ;
        TITLE2 &title2 ;
```

```
run ;

proc sgplot data = &dsout. noautolegend ;

        scatter y=&yvar x=&xvar / MARKERATTRS=(SIZE=5px)
                                name="Assay1" ; *COLOR=blue SYMBOL=triangle) ;

        vector x=Lx y=Ly /xorigin=&x0 yorigin=&y0 noarrowheads
                                        LINEATTRS = (THICKNESS=0.1 COLOR=Blue PATTERN=1) ;
        LINEPARM x=0 y=0 slope=1 / LEGENDLABEL = 'Unity: Y=X'
         LINEATTRS = (THICKNESS=0.1 COLOR=Black PATTERN=2) name="Unity" ; * works in SAS v9.3 ;

%if &GraphMin. eq  and &GraphMax. eq
       %then %do ;
                YAXIS LABEL = &ylab. ;
                XAXIS LABEL = &xlab. ;
          %end ;
         %else %do ;
                YAXIS LABEL = &ylab.
                  VALUES = (
                            %if &GraphMin. = %then
                                %do ;
                                  &GraphMin1.
                                %end ;
                             %else %do ;
                                   &GraphMin.
                              %end ;
                              TO
                             %if &GraphMax. = %then
                                 %do ;
                                    &GraphMax1.
                                 %end ;
                                 %else %do ;
                                    &GraphMax.
                                 %end ;
                                 BY &GraphIncrement.) ;
                XAXIS LABEL = &xlab.
                  VALUES = (
                            %if &GraphMin. = %then
                                %do ;
                                    &GraphMin1.
                                %end ;
                                %else %do ;
                                    &GraphMin.
                                %end ;
                                TO
                            %if &GraphMax. = %then
                                %do ;
                                    &GraphMax1.
                                %end ;
                                %else %do ;
                                    &GraphMax.
                                %end ;
                                BY &GraphIncrement.
                            ) ;
         %end ;

        INSET "Deming Regression (N=&n): Y = &intercept + &slope X"
        "95% CIs: Intercept=&interceptci ; Slope=&slopeci" / POSITION = BOTTOMRIGHT BORDER ;

        TITLE1 &title1 ;
        TITLE2 &title2 ;
run ;
%MEND demreg ;
* END OF MACRO CODE ;
```

APPENDIX B: DEMREG SAS MACRO CALL FOR EXAMPLE 1

```
%LET TodaysDate = %sysfunc(TODAY(),DATE9.) ;
%LET Path  = c:/pharmasug/2022_Austin ;
%LET SEED1 = 726453819 ;
%LET SEED2 = 289546394 ;
%LET SEED3 = 923847362 ;
ods listing close;
data Example ;
 do MyId = 1 to 200 ;
  Value = UNIFORM(&SEED1.) * 10 ;
  Ref = Value + NORMAL(&SEED2.) ;
  Test = Value + NORMAL(&SEED3.) ;
  output ;
 end ;
 drop Value ;
run ;


* Use the MODSTYLE macro supplied by SAS to change the default colors and markers ;
%modstyle(name        = markers ,
          parent      = listing ,
          type        = CLM     ,
          markers     = circle triangle square            diamond      star   ,
          colors      = blue   green    red               purple       orange ,
          linestyles = Solid  LongDash MediumDashDotDot  ShortDashDot ThinDot) ;

* If the investigators want a square plot (no proc template necessary) ;
ods _all_ close ;
ods listing;
ods graphics on;
ods listing gpath="&path./results/png" ;
ods graphics / width = 550px height = 550px noborder outputfmt=PNG ;

%include "&path./macros/demreg.sas" ;
ods noproctitle ;
ods rtf file = &path./results/rtf/DemReg_Example_&TodaysDate..rtf" gtitle style=markers ;
%demreg(dsin=example
,       dsout=example_out
,       title1="Method: Wicklin (2019)(lambda=1)"
,       title2="Deal et al. example"
,       sampleid= MyID
,       xvar= ref
,       xlab= "reference"
,       yvar= test
,       ylab= "test"
,       alpha= 0.05
,      lambda= 1
,      GraphMin= -3
,      GraphMax= 13
,      GraphIncrement= 1
,      ParmDecimalPlaces= 0.001 ) ;
ods rtf close ;
```

*** END OF MACRO CALL ;**

□